OS Development

Luke

December 11, 2024

Abstract

This document serves as a reference for the Minimus operating system.

Contents

1 Bootloader

1.1 Headers

The disk is ordered into sectors, of size 200_{16}, starting from sector 1[3]. When the BIOS loads the OS, it copies the first sector, which would be the first 512 bytes, from the disk into memory, starting at $7C00_{16}$[1].

Memory is also ordered into segments, of size 10_{16}[4]. This means that memory addresses can overlap, for example: The address $0000{:}7C00_{16}$ is the same as $0700{:}0C00_{16}$.

You must ensure that the bytes at $1FE_{16}$ and $1FF_{16}$ contain values 55_{16} and AA_{16} respectively[5]. This is called the magic number, and is used to differentiate bootable disks from non-bootable disks.

When the BIOS hands control to the OS, the CPU will be in 16 bit (real) mode[6], which means it is using 16 bits per instruction. This is because the CPU is in 16 bit mode by default. You must ensure that your program code for the bootloader begins in 16 bit (assuming real) mode.

`bootloader/main.asm`

```
1 [org 0x7c00]                  ; memory load location
2 [bits 16]                   ; real mode
```

1.2 Disk read

Reading from the disk is done with Bios Interrupt 13_{16} ah=02[2].

The interrupt specification is layed out below.1:

Register	Value
AH	02
AL	Number of sectors to read
CH	Cylinder
DH	Head
CL	Sector
DL	Drive
ES:BX	Output offset

Since the DL register is initialised with the boot drive before control is handed to our program[7], as long as it is not overwritten before calling int 13_{16}, you do not have to alter it.

Since the ES register cannot be written to directly, due to no CPU instruction being availible to transfer a value to ES[8], which results in an intemediary value needing to be used.

I decided to use BX, due to it being overwritten used in the instruction after, but any unused register will do. Then, I assigned BX with $7e00_{16}$, which

is 200_{16} bytes more than the start of the program[1], which is the precise number of bytes loaded by the BIOS[3].

The number of load segments (register AX)[2] have a limit of around 100 on some systems, and around 70 on QEMU. 64 is a safe number of sectors to load, and allows for 32KB of kernel instructions (which is far more than enough).

Sometimes there will be an error, stored in the carry bit[2], if this happens, retry the load from disk. An error may also be indicated by the AL register having an incorrect number of sectors read.

`bootloader/main.asm`

```
KERNEL_SEGS equ 64        ; KERNEL_SEGS * 512

; read kernel (https://en.wikipedia.org/wiki/INT_13H)
mov bx, 0
mov es, bx
mov bx, 0x7e00              ; offset
mov ah, 0x02               ; set read mode
mov cl, 2                 ; start from sec 2
mov al, KERNEL_SEGS         ; sectors to read
mov ch, 0                 ; cylinder
mov dh, 0                 ; head
int 0x13                 ; call
jc $$                       ; carry bit stores error, loop
cmp al, KERNEL_SEGS        ; al is sectors read
jne $$                       ; if all sectors arent read, loop
```

1.3 Usable memory

The amount of usable memory varies between systems, and some memory is reserved for hardware, VGA and VBE, console video memory, etc.

Detecting availible memory blocks is done with interrupt 15_{16}[12].3. Although you can probably get away with ignoring this, and just hoping memory above, lets say 1mb, is completely fine, I went through the extra effort to correctly get usable memory.

This area repeatedly calls int 15_{16}, and saves the result, if it is larger than the currently tested memory, in a memory location ironically not tested to see if it is reserved.

The copying and checking is done in two parts due to the fact that the numbers are larger than the 16 bit memory availible in real mode.

`bootloader/main.asm`

```
; get largest availible memory block
pusha                       ; push all
mov cx, 0x0                ; clear cx for addition later
```

```
xor eax, eax
mov es, eax
mov ebx, 0x0                ; clear
mov edx, 0x534d4150         ; magic value
memreadloop:
        mov eax, 0xe820             ; magic value
        mov ecx, 0x18             ; magic value
        mov di, 0x7bd0             ; memory location for buffer
        int 0x15             ; call function
        add di, cx    ; increment di by entry size
                      ; (cx is 16 bit cl)
memreadloopvalid:
        mov eax, [0x7be0]    ; load type
        cmp eax, 1            ; check if 1 (availible memory)
        jne memreadloopend    ; go to next otherwise
memreadloopcheck:
        mov eax, [0x7bda]        ; load size of current
        mov ecx, [0x7bfa]        ; load size of biggest
        cmp eax, ecx               ; check if bigger
        jle memreadloopend        ; go to next otherwise
memreadlooprecord:
        mov eax, [0x7bd0]        ; load address of current
        mov [0x7bf0], eax        ; record biggest address
        mov eax, [0x7bd4]        ; load address of current
        mov [0x7bf4], eax        ; record biggest address
        mov eax, [0x7bd8]        ; load size of current
        mov [0x7bf8], eax        ; record biggest size
        mov eax, [0x7bdc]        ; load size of current
        mov [0x7bfc], eax        ; record biggest size
memreadloopend:
        cmp ebx, 0                ; check if next
        jnz memreadloop                ; repeat
```

1.4 Display

1.4.1 VGA

Enabling VGA[14] here allows me to get the font later, and also serves as compatibility for if VESA is not supported.

The options for AL are defined.4.

`bootloader/main.asm`

```
; vga mode
mov ah, 0x0                ; graphics mode
```

```
3  mov al, 0x13                ; 256 colour 200x320
4  int 0x10                  ; set vga mode
```

1.4.2 Font

The code below gets the font from the VGA card. This is so I did not have to store my own.

The code is taken mostly from this OsDev page[13], due to the fact that I figured it was mostly copy paste anyway.

`bootloader/main.asm`

```
1  ; get vga font
2  mov eax, 0x100
3  mov es, eax
4  mov ax, 0x0
5  mov di, ax
6  push ds
7  push es
8  mov ax, 0x1130              ; magic numbers
9  mov bh, 0x6
10 int 0x10                  ; get vga font
11 ;mov ds, es
12 push es
13 pop ds
14 pop es
15 mov si, bp
16 mov cx, 0x400
17 rep movsd
18 pop ds
```

1.4.3 VBE

The VESA bios extensions allows for greater colour depth (RGB) and higher resolutions than VGA. Getting the VBE table is done with interrupt 10_{16}[15].

Once sifted to find the right function, it can again be enabled with interrupt 10_{16}, the magic numbers for each function are in the VBE spec[15].

The resulting values (whether 24 bit or 32 bit colour is used, the location of the framebuffer, whether VBE is even supported in the first place, etc), are placed into another (not checked) area of memory, which will be used by the kernel later.

I start by checking for VBE2 support, and if it doesn't exist I jump to the end of the VBE section. Then I go through each function, checking if it has RGB colourspace, and is the correct resolution. I then update the bytes at some position (again not fully checked) so my kernel has access to whether VBE was

enabled or not, and also to provide the kernel with the position of the linear framebuffer.

This is due to the fact that in VBE2, it is not required that all the VBE1 functions.5 are supported (even though they most likely are).

Enabling a VBE mode is as simple as VBE once the functions have been sorted, and is the same BIOS interrupt.6.

`bootloader/main.asm`

```
; get vesa support
mov [0x2000], DWORD "VBE2"
mov ax, 0x4f00                  ; magic number
mov di, 0
mov es, di
mov di, 0x2000                  ; offset to table
int 0x10                ; get vbe table
cmp ax, 0x004f                  ; check if support
jne skip_vbe                ; use vga instead if not

mov di, 0                ; offset to tmp
mov es, di
mov di, 0x2200                  ; segment to tmp
mov si, [0x2000 + 16]         ; segment of list
mov ds, si
mov si, [0x2000 + 14]         ; offset of list

vbe_loop:
; get next supported vesa function
movsw
sub di, 2                ; (2 is added to both)
mov cx, [0x2200]         ; vesa mode
cmp cx, 0xffff                  ; check if end
je skip_vbe               ; loop

; get details
mov ax, 0x4f01                  ; magic number
int 0x10                ; get vbe

mov ax, [0x2200]         ; check attr
and ax, 0b10000000         ; check bit 7 (linear framebuffer)
cmp ax, 0
jz vbe_loop               ; if not linear, loop

mov al, [0x2200 + 25]         ; check colourspace
mov [0x2201], BYTE 0x00         ; vbe 32 bit flag
cmp al, 24
```

```
38  je vbe_check_dims        ; if not rgb, loop
39  mov [0x2201], BYTE 0xff        ; vbe 32 bit flag
40  cmp al, 32
41  je vbe_check_dims        ; if not rgb, loop
42  jmp vbe_loop               ; actual loop
43
44  vbe_check_dims:
45  mov ax, [0x2200 + 18]        ; check width
46  cmp ax, 640
47  jne vbe_loop               ; if not desired, loop
48
49  mov ax, [0x2200 + 20]        ; check height
50  cmp ax, 480
51  jne vbe_loop               ; if not desired, loop
52
53  jmp enable_vbe               ; turn on this vbe
54
55  skip_vbe:
56  mov [0x2200], BYTE 0x00        ; vbe corect flag
57  jmp complete_vbe        ; skip
58
59  enable_vbe:
60  mov ax, [0x2200 + 40]        ; framebuffer
61  mov [0x2200 + 2], ax        ; framebuffer
62  mov ax, [0x2200 + 42]        ; framebuffer
63  mov [0x2200 + 4], ax        ; framebuffer
64  mov ax, 0x4f02               ; magic number
65  mov bx, cx                ; move mode number
66  or bx, 0x4000               ; set linear framebuffer
67  int 0x10                ; set vbe mode
68  cmp ax, 0x004f               ; test for error
69  jne skip_vbe               ; skip if error (will use vga
    ↪ instead)
70  mov [0x2200], BYTE 0xff        ; vbe correct flag
71
72  complete_vbe:
73  popa                       ; pop all
```

1.5 Enabling 32-bit processing

1.5.1 Segment descriptor

The Global Descriptor Table is necessary for enabling protected mode[9], and is defined below in a basic form in order to switch to protected mode.

For my kernel, none of the features of the GDT, like paging[11], are not needed, so the GDT is very basic.2.

bootloader/main.asm

```
jmp gdt_after

; segment descriptor (reverse order)
gdt_start:
        dq 0                ; null byte start
gdt_code:
        dw 0xffff         ; segment limit
        db 0,0,0         ; segment base
        db 0b10011010          ; flags (see wiki)
        db 0b11001111          ; 4b flags (see wiki) + seg limit
        db 0                 ; segment base
gdt_data:
        dw 0xffff         ; segment limit
        db 0,0,0         ; segment base
        db 0b10010010          ; flags (see wiki)
        db 0b11001111          ; 4b flags (see wiki) + seg limit
        db 0                 ; segment base
gdt_end:
        dw gdt_end - gdt_start - 1        ; limit
        dd gdt_start                        ; addr 24 bit
gdt_after:
```

1.5.2 Prerequisites

To enable protected mode, you must first disable interrupts, and load the global descriptor table[16], which can be done in NASM with commands `cli` and `lgdt [GDT ADDRESS]` respectively.

bootloader/main.asm

```
cli                         ; disable interrupts
lgdt [gdt_end]                ; gdt_end is descritor table
↪  descriptor
```

1.5.3 Protected mode

Upon the completion of all the above, you can then set bit 0 (starting from 0) to 1 in the control register CR0[16], which must be done in seperate commands, as the special register CR0 is not directly assignable[8].

You must then immediately[16] perform a long jump to your next desired instruction.

You then have to tell the assembler that you are now using 32 bits, which can be done with the `[bits 32]` command in NASM.

bootloader/main.asm

```
mov eax, cr0
or eax, 1                ; set 1 bit in control register for protected mode
mov cr0, eax
jmp (gdt_code - gdt_start):bits32code        ; stall cpu and flush all cache (as moving to different segment) to finalize protected mode

; finally 32 bits
[bits 32]
bits32code:
```

1.6 Enabling the A20 line

Enabling the A20 line (the proper way) is by the keyboard[10]. This is incredibly tedious and repetitive, which is why I have taken this code from the OsDev page.

Two small assembly functions: a20wait; a20waitr, have been created to avoid some of the repetition. a20wait will repeatedly poll port 64_{16} until bit 1 is set (which is why `TEST` is used). a20waitr will do the same until bit 0 is set.

This is due to the fact that it is required to wait for bit 1 in port 64_{16} before writing and bit 0 in port 64_{16} before reading.

A brief outline of the functions of the A20 line are provided within this document.7.

bootloader/main.asm

```
xor al, al
call a20wait                ; wait for write
mov al, 0xad
out 0x64, al                ; send 0xad
call a20wait                ; wait for write
mov al, 0xd0
out 0x64, al                ; send 0xd0
call a20waitr                ; wait for read
in al, 0x60                ; get ack
push eax
call a20wait                ; wait for write
mov al, 0xd1
out 0x64, al                ; send 0xd1
call a20wait                ; wait for write
pop eax                        ; eax gets overwritten
or al, 0b0010                ; a20 bit
out 0x60, al                ; set a20 bit on
call a20wait                ; wait for write
```

```
mov al, 0xae
out 0x64, al                ; send 0xae
call a20wait                ; wait for generic
jmp skipa20                 ; go to end
a20wait:
        in al, 0x64
        test al,2
        jnz a20wait
        ret
a20waitr:
        in al, 0x64
        test al,1
        jz a20waitr
        ret

skipa20:
```

1.7 Loading the Kernel

Before loading the kernel, I initialise all the registers high parts with the segment and move the stack pointers to 7000_{16}.

The bootloader begins at $7c00_{16}$[1], and because the stack grows downwards, 7000_{16} is a perfect location for a stack. You can place yours where you like.

`bootloader/main.asm`

```
jmp (gdt_code - gdt_start):start_kernel

start_kernel:
        ; segment registers init
        mov ax, gdt_data - gdt_start
        mov ds, ax
        mov es, ax
        mov fs, ax
        mov gs, ax
        mov ss, ax

        ; stack pointers
        mov esp, 0x7000             ; top of stack
        mov ebp, esp                ; bottom of stack

        call kernel                 ; start kernel and move back
        ↪  to segment

; kernel
kernel:
```

```
20          jmp kernel_loadseg          ; hand control to kernel
21          jmp $$                          ; return -> error, loop
22
23                              ...
24
25  ; kernel load
26  kernel_loadseg:
27  call kernel_cseg
28  jmp $                                   ; if fail restart
29  kernel_cseg:                           ; compiled c appeneded here
```

1.8 The boot signature

You must ensure that the bytes at $1FE_{16}$ and $1FF_{16}$ contain values 55_{16} and AA_{16} respectively[5]. This is called the boot signature, or magic number, and is used to differentiate bootable disks from non-bootable disks, and also tells the BIOS how to load your OS.

`bootloader/main.asm`

```
1  ; padding
2  times 510 - ($-$$) db 0
3
4  ; boot signature
5  db 0x55,0xaa
```

2 Appendix

.1 INT 13_{16} AH 02_{16}

Register	Value
AL	Number of sectors to read
CH	Cylinder
DH	Head
CL	Sector
DL	Drive
ES:BX	Output offset

.2 Segment Descriptor

Bits	Segment	Value
0-15	*(1)*	Segment limit
16-39	*(1)*	Segment base
40-43	*(flag)*	Type
44	*(flag)*	S
45-46	*(flag)*	DPL
47	*(flag)*	P
48-51	*(2)*	Segment limit
52	*(flag)*	A
53	*(null)*	0
54	*(flag)*	DB
55	*(flag)*	G
56-63	*(2)*	Segment base

.3 INT 15_{16} EAX $E820_{16}$

Register	Value
EAX	$E820_{16}$
EBX	0
ECX	24
EDX	$534D4150_{16}$
ES:DI	Output buffer

.4 VGA Options

Value	Type	Resolution	Colourspace
00_{16}	Text	40x25	1
01_{16}	Text	40x25	16
02_{16}	Text	80x25	1
03_{16}	Text	80x25	16
04_{16}	CGA	320x200	4
05_{16}	CGA	320x200	1
06_{16}	CGA	640x200	2
07_{16}	MDA	80x25	1
$0D_{16}$	EGA	320x200	16
$0E_{16}$	EGA	640x200	16
$0F_{16}$	EGA	640x350	1
10_{16}	EGA	640x350	16
11_{16}	VGA	640x480	1
12_{16}	VGA	640x480	16
13_{16}	VGA	320x200	256

.5 VBE1 Functions

Value	Type	Resolution	Colourspace
100_{16}	Graphics	640x400	256
101_{16}	Graphics	640x480	256
103_{16}	Graphics	800x600	256
104_{16}	Graphics	1024x768	16
105_{16}	Graphics	1024x768	256
106_{16}	Graphics	1280x1024	16
107_{16}	Graphics	1280x1024	256
$10D_{16}$	Graphics	320x200	32K (1:5:5:5)
$10E_{16}$	Graphics	320x200	64K (5:6:5)
$10F_{16}$	Graphics	320x200	16.8M (8:8:8)
110_{16}	Graphics	640x480	32K (1:5:5:5)
111_{16}	Graphics	640x480	64K (5:6:5)
112_{16}	Graphics	640x480	16.8M (8:8:8)
113_{16}	Graphics	800x600	32K (1:5:5:5)
114_{16}	Graphics	800x600	64K (5:6:5)
115_{16}	Graphics	800x600	16.8M (8:8:8)
116_{16}	Graphics	1024x768	32K (1:5:5:5)
117_{16}	Graphics	1024x768	64K (5:6:5)
118_{16}	Graphics	1024x768	16.8M (8:8:8)
119_{16}	Graphics	1280x1024	32K (1:5:5:5)
$11A_{16}$	Graphics	1280x1024	64K (5:6:5)
$11B_{16}$	Graphics	1280x1024	16.8M (8:8:8)

.6 INT 10_{16} AX $4F02_{16}$

Register	Value
AX	$4F02_{16}$
BX	Mode
BX	$4XXX_{16}$ for linear framebuffer

.7 Enabling A20 Line through the Keyboard Controller

Port	Value
64_{16}	AD_{16}
64_{16}	$D0_{16}$
60_{16}	*read*
64_{16}	$D1_{16}$
60_{16}	*(read)* $_{OR}$ *2*
64_{16}	AE_{16}

References

[1] Compaq Computer Corporation, Phoenix Technologies Ltd, Intel Corporation (1996) *BIOS Boot Specification* pg 29 ch 6.5.1

[2] Ralf Brown (2000) *Ralf Browns Interrupt List* interrup b int 13

[3] IDEMA (2013) *The Advent of Advanced Format* pg 1

[4] Intel Corporation (2016) *Intel 64 and IA-32 Architectures Software Developer's Manual Volume 1: Basic Architecture* pg 1.6 ch 1.3.4

[5] Compaq Computer Corporation, Phoenix Technologies Ltd, Intel Corporation (1996) *BIOS Boot Specification* pg 12 ch 3.3

[6] Intel Corporation (2016) *Intel 64 and IA-32 Architectures Software Developer's Manual Volume 3A: System Programming Guide, Part 1* pg 11.13 ch 1.9.1

[7] Compaq Computer Corporation, Phoenix Technologies Ltd, Intel Corporation (1996) *BIOS Boot Specification* pg 43 ch D.1

[8] Intel Corporation (2022) *Intel 64 and IA-32 Architectures Software Developer's Manual Volume 2 (2A, 2B, 2C, 2D): Instruction Set Reference, A-Z* pg 4.35 ch 3.3

[9] Wikipedia (2024) *Global Descriptor Table* ch Description

[10] Robert Collins (2001) *A20/Reset Anomalies* ch A20/Reset Anomalies

[11] Wikipedia (2024) *Global Descriptor Table* ch Modern Usage

[12] OsDev (2024) *Detecting Memory (x86)* ch 2.1

[13] OsDev (2024) *VGA Fonts* ch 2.4

[14] Wikipedia (2024) *Mode 13h* ch 2.4

[15] Video Electronics Standards Association (1998) *VESA BIOS EXTENSION (VBE) Core Functions Standard* pg 17

[16] Intel Corporation (2016) *Intel 64 and IA-32 Architectures Software Developer's Manual Volume 3A: System Programming Guide, Part 1* pg 9.17 ch 9.9.2